<!DOCTYPE HTML PUBLIC "-//W3C//DTD HTML 4.01 Transitional//EN">
<html><head>
  
  <meta http-equiv="Content-Type" content="text/html; charset=utf-8">

  
  <link href="../css/style.css" rel="stylesheet" type="text/css">

</head>
<body>
<h1>4.18. Синтаксис объявления меток</h1>
<p class="article">
Синтаксис FASM не ограничивается объявлением простых меток. В этой части мы рассмотрим дополнительную директиву для создания меток, а также научимся использовать локальные и анонимные метки.
В синтаксисе FASM существует 3 основных способа объявления меток:
</p>
<p class="article">
1. Имя метки, после которого ставится двоеточие. Это самый простой способ. Обычно так объявляются метки в коде.
</p>
<pre class="code">
exit_app:
    mov ax,4C00h
    int 21h
</pre>
<p class="article">
2. Использование директив объявления данных. Имя переменной является по сути той же меткой. Отличие от первого способа в том, что дополнительно с именем метки связывается размер переменной. 
</p>
<pre class="code">
x db 5
y dw 34,1200,?
z rd 1
</pre>
<p class="article">
3. Объявление метки с помощью специальной директивы label. Более сложный, но зато самый гибкий способ. Его мы рассмотрим подробнее.
Директива label имеет следующий формат:
</p>
<pre class="code">
label <имя_метки> [размер] [at адрес]
</pre>
<p class="article">
У директивы может быть 3 параметра. Обязательным является только первый параметр — имя метки. Второй параметр — оператор размера (byte, word, dword и т.д.). Он связывает с меткой размер переменной, аналогично тому, как это делают директивы объявления данных. Далее может быть указан оператор at и адрес метки. Адрес может представлять собой константу, числовое выражение или имя другой метки. Если адрес не указан, то для создания метки используется адрес того места, где она объявлена.
</p>
<pre class="code">
label m1              ;То же самое, что 'm1:'
label m2 byte         ;Похоже на 'm2 db ?', но память не резервируется
label m3 dword        ;Похоже на 'm3 dd ?', но память не резервируется
                      ;Все 3 метки указывают на один и тот же адрес!
</pre>
<p class="article">
Следующий пример показывает, как можно использовать в программе директиву label. Допустим, объявлена переменная x размером слово. Требуется обнулить старший байт x. Воспользовавшись директивой label, можно обратиться к старшему байту как к отдельной переменной:
</p>
<pre class="code">
x   dw 12345          ;Переменная-слово
label xh byte at x+1  ;Объявление метки для обращения к старшему байту
...
start:
    xor al,al         ;AL=0
    mov [xh],al       ;xh=0 (старший байт x)
    mov byte[x+1],al  ;То же самое без использования метки xh
</pre>
<p class="article">
Кроме того, адрес может содержать базовые и индексные регистры для косвенной адресации. Такие метки можно использовать для обращения к параметрам и локальным переменным процедуры. Например:
</p>
<pre class="code">
abel i word at bp-2  ;Локальная переменная
    ...
    inc [i]           ;Инкремент локальной переменной
</pre>
<h2>Локальные метки</h2>
<p class="article">
Локальная метка — это метка, имя которой начинается с точки. Во время генерации кода FASM автоматически добавляет к имени локальной метки имя последней объявленной «глобальной» метки. Таким образом, имена локальных меток могут повторяться, если между ними есть хотя бы одна «глобальная» метка.
Локальные метки удобно использовать, например, внутри процедуры. Можно дать им простые, понятные имена и не беспокоиться, что где-то в коде уже объявлена метка с таким именем.
</p>
<div class="notetip">
Локальные метки намного улучшают читаемость кода. Если потребуется обратиться к локальной метке из другого места программы, это можно сделать, указав её полное имя:
<pre class="code">
    jmp str_to_udec_word.error  ;Переход к локальной метке
</pre>
</div>
<h2>Анонимные метки</h2>
<p class="article">
Анонимная метка — это метка с именем @@. В программе можно объявлять сколько угодно анонимных меток, но обратиться получится только к ближайшей. Для этого существуют специальные имена: вместо @b (или @r) FASM подставляет адрес предыдущей анонимной метки, а вместо @f — адрес следующей анонимной метки. Этого, как правило, достаточно, чтобы реализовать простой цикл, переход или проверку условия. Таким образом можно избавиться от большого количества «неанонимных» меток. Вот пример той же процедуры с использованием анонимных меток:
</p>
<pre class="code">
;Процедура преобразования десятичной строки в слово без знака
;  вход: AL - длина строки
;        DX - адрес строки, заканчивающейся символом CR(0Dh)
; выход: AX - слово (в случае ошибки AX = 0)
;        CF = 1 - ошибка
str_to_udec_word:
    push cx                 ;Сохранение всех используемых регистров
    push dx
    push bx
    push si
    push di
 
    mov si,dx               ;SI = адрес строки
    mov di,10               ;DI = множитель 10 (основание системы счисления)
    movzx cx,al             ;CX = счётчик цикла = длина строки
    jcxz .error             ;Если длина = 0, возвращаем ошибку
    xor ax,ax               ;AX = 0
    xor bx,bx               ;BX = 0
 
@@: mov bl,[si]             ;Загрузка в BL очередного символа строки
    inc si                  ;Инкремент адреса
    cmp bl,'0'              ;Если код символа меньше кода '0'
    jl .error               ; возвращаем ошибку
    cmp bl,'9'              ;Если код символа больше кода '9'
    jg .error               ; возвращаем ошибку
    sub bl,'0'              ;Преобразование символа-цифры в число
    mul di                  ;AX = AX * 10
    jc .error               ;Если результат больше 16 бит - ошибка
    add ax,bx               ;Прибавляем цифру
    jc .error               ;Если переполнение - ошибка
    loop @b                 ;Команда цикла
    jmp @f                  ;Успешное завершение (здесь всегда CF = 0)
 
.error:
    xor ax,ax               ;AX = 0
    stc                     ;CF = 1 (Возвращаем ошибку)
 
@@: pop di                  ;Восстановление регистров
    pop si
    pop bx
    pop dx
    pop cx
    ret
</pre>
</body>
</html>
